


# Film Slitting Machine

## [System Configuration]



## [Mitsubishi solution]

Motion CPU: Q172DSCPU  
PLC CPU: Q06UDEHCPU  
Main base: Q35DB

Engineering environment: MELSOFT MT Works2 (Motion), MELSOFT GX Works2 (PLC), MELSOFT GT Works3 (GOT)  
Motion CPU operating system software: SW8DNC-SV22QL

GOT: GT27\*\*-V  
Analog input module: Q64AD

Servo amplifier: MR-J4-B  
Servo motor: HG-SR


## [Operation description]

In a system comprising an unwinding axis, feed axis, cutter axis, and rewinding axis, a material such as film is unwound, slit by cutter, and rewound.

## [Control points]

- Point1: The unwinding axis outputs a speed command depending on the film outer circumference. The film is unwound at a constant tension by correcting the speed command value so that the feedback value from the tension detector equals the intended tension.
- Point2: The rewinding axis outputs a speed command depending on the film diameter. The film is rewound at a constant tension by correcting the speed command value so that the dancer roll stays in a fixed position.
- Point3: Speed control is executed using advanced synchronous control to synchronize the entire line.
- Point4: Energy saving is promoted by utilizing the regenerative energy on the unwinding axis as power running energy for the rewinding axis using the common converter.

## [Typical Machinery Configuration]




(Note): Controls for preparation before operation start are not included in this sample program.

Add following controls depend on the actual system.

- Initial film feed operation
- Initial film outer circumference/diameter measurement
- Initial tension adjustment and initial dancer roll position adjustment

## [Control block diagram]




## [GOT Sample Screen]

[GOT: Home Screen]




[GOT: Main Screen]



(Note): Sample screen as default are set for English environment. When using Japanese environment, it's possible to switch to Japanese for GOT monitoring data in GT Designer 3 Language change the preview column from [2] to [1].

## [Parameter Setting]

### (1) System Setting



## (2) Servo data setting

### Electronic gear setting

Movement amount per revolution: 200[mm] (Roller outer circumference: 600[mm], Gear ratio: 1/3)

Item	Axis1[Unwinding]	Axis2[Feed]	Axis3[Cutter]	Axis4[Rewinding 1]	Axis5[Rewinding 2]
<b>Set the fixed parameters for each axis and their data is fixed based on the mechanical system, etc.</b>					
Unit Setting	0:mm	0:mm	0:mm	0:mm	0:mm
Number of Pulses/Rev.	4194304[PLS]	4194304[PLS]	4194304[PLS]	4194304[PLS]	4194304[PLS]
Travel Value/Rev.	200000.0[µm]	200000.0[µm]	200000.0[µm]	200000.0[µm]	200000.0[µm]
Backlash Compensation	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]
Upper Stroke Limit	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]
Lower Stroke Limit	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]
Command In-position	10.0[µm]	10.0[µm]	10.0[µm]	10.0[µm]	10.0[µm]
Sp. Ctrl. 10x Mult. for Deg.	-	-	-	-	-
<b>Set the data to execute the home position return.</b>					
OPR Direction	0:Reverse Direction	0:Reverse Direction	0:Reverse Direction	0:Reverse Direction	0:Reverse Direction
OPR Method	2:Data Set Type 1				
Home Position Address	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]	0.0[µm]
Operation for OPR Incompletion	0:Execute Servo Program				
<b>Set the data to execute the JOG operation.</b>					
JOG Speed Limit Value	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]
Parameter Block Setting	1	1	1	1	1
<b>Set the data only when the speed-torque control is executed.</b>					
Control Mode Switching Request Device	M1	M2	M3	M4	M5
Control Mode Setting Device	D1000(1)	D1010(1)	D1020(1)	D1030(1)	D1040(1)
Speed Limit Value in Speed-Torque Control	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]	600000.00[mm/min]
Torque Limit Value in Speed-Torque Control	300.0[%]	300.0[%]	300.0[%]	300.0[%]	300.0[%]
Speed Command Device	D1002(2)	D1012(2)	D1022(2)	D1032(2)	D1042(2)
Command Speed Acceleration Time	1000[ms]	1000[ms]	1000[ms]	1000[ms]	1000[ms]
Command Speed Deceleration Time	1000[ms]	1000[ms]	1000[ms]	1000[ms]	1000[ms]
Torque Command Device	D1004(1)	D1014(1)	D1024(1)	D1034(1)	D1044(1)
Command Torque Time Constant (Positive Dire...	1000[ms]	1000[ms]	1000[ms]	1000[ms]	1000[ms]
Command Torque Time Constant (Negative Dire...	1000[ms]	1000[ms]	1000[ms]	1000[ms]	1000[ms]
Speed Initial Value Selection at Control Mo...	0:Command Speed				
Torque Initial Value Selection at Control Mo...	0:Command Torque				
Invalid Selection during Zero Speed at Control ...	0:Switching Condition at Switching Control ...	0:Switching Condition at Switching Control ...	0:Switching Condition at Switching Control ...	0:Switching Condition at Switching Control ...	0:Switching Condition at Switching Control ...


### Control mode switching setting

Set the control data for position/speed/torque/continuous operation to torque control mode. For this sample program, the mode is switched to speed mode on all axes at synchronous operation.

<Speed control using advanced synchronous control>

In advanced synchronous control, the speed command to output axis is equal to the speed command in speed control mode.  
(Speed control by speed command device is not executed.)

Control mode can be switched in the linear cam (cam No.0) only.




Command speed at speed control mode

### (3) Synchronous control parameter

#### [Command generation axis parameters]

Item	Axis 1
<b>Command Generation Axis</b>	
Valid Setting	1:Valid
Unit Setting	0:mm
Upper Stroke Limit	0.0 µm
Lower Stroke Limit	0.0 µm
Command In-position Range	10.0 µm
Sp. Ctrl. 10x Mult. for Deg.	-
Length per Cycle	600000.0 µm
JOG Speed Limit Value	600000.00 mm/min
JOG Operation Parameter Block Setting	1
<b>Acceleration/deceleration Time Change Parameter</b>	<b>Set acceleration/decel</b>

#### [Synchronous parameter]



#### **Cautions**

- When diverting the sample program to the actual system, be sure to verify that there are no problems with control in the system.
- Add interlock conditions in the target system where considered necessary.

#### [Sample program configuration]

File name	Description	Model	Engineering environment
Vol9_Converting_Motion_b.mtw	Motion control program Motion parameter	Q172DSCPU	MELSOFT MT Works2
Vol9_Converting_GOT_b.GTX	GOT monitoring data	GT27**-V (640x480)	MELSOFT GT Works3

## [Devices used in this program]

Q172DSCPU dedicated devices

Device No.	Content	Device No.	Content
M2001	Axis 1 start accept flag	M9810	Command generation axis 1 start accept flag
M2002	Axis 2 start accept flag	M10880	Axis 1 synchronous control
M2003	Axis 3 start accept flag	M10881	Axis 2 synchronous control
M2004	Axis 4 start accept flag	M10882	Axis 3 synchronous control
M2005	Axis 5 start accept flag	M10883	Axis 4 synchronous control
M2042	All axis servo ON	M10884	Axis 5 synchronous control
M2407	Axis 1 error detection	M10960	Command generation axis 1 stop
M2408	Axis 1 servo error detection	M12000	Axis 1 synchronous control start
M2410	Axis 1 home position return completion	M12001	Axis 2 synchronous control start
M2415	Axis 1 servo READY	M12002	Axis 3 synchronous control start
M2427	Axis 2 error detection	M12003	Axis 4 synchronous control start
M2428	Axis 2 servo error detection	M12004	Axis 5 synchronous control start
M2430	Axis 2 home position return completion	D2	Axis 1 actual current value
M2435	Axis 2 servo READY	D3	
M2447	Axis 3 error detection	D22	Axis 2 actual current value
M2448	Axis 3 servo error detection	D23	
M2450	Axis 3 home position return completion	D62	Axis 4 actual current value
M2455	Axis 3 servo READY	D63	
M2467	Axis 4 error detection	D82	Axis 5 actual current value
M2468	Axis 4 servo error detection	D83	
M2470	Axis 4 home position return completion	D640	Axis 1 JOG speed setting
M2475	Axis 4 servo READY	D641	
M2487	Axis 5 error detection	D642	Axis 2 JOG speed setting
M2488	Axis 5 servo error detection	D643	
M2490	Axis 5 home position return completion	D644	Axis 3 JOG speed setting
M2495	Axis 5 servo READY	D645	
M3202	Axis 1 forward JOG start	D646	Axis 4 JOG speed setting
M3203	Axis 1 reverse JOG start	D647	
M3207	Axis 1 error reset	D648	Axis 5 JOG speed setting
M3208	Axis 1 servo error reset	D649	
M3222	Axis 2 forward JOG start	D15050	Axis 1 speed change gear ratio 1 denominator
M3223	Axis 2 reverse JOG start	D15051	
M3227	Axis 2 error reset	D15054	Axis 2 speed change gear ratio 2 numerator
M3228	Axis 2 servo error reset	D15055	
M3242	Axis 3 forward JOG start	D15498	Axis 4 speed change gear ratio 1 numerator
M3243	Axis 3 reverse JOG start	D15499	
M3247	Axis 3 error reset	D15500	Axis 4 speed change gear ratio 1 denominator
M3248	Axis 3 servo error reset	D15501	
M3262	Axis 4 forward JOG start	D15504	Axis 4 speed change gear ratio 2 numerator
M3263	Axis 4 reverse JOG start	D15505	
M3267	Axis 4 error reset	D15506	Axis 4 speed change gear ratio 2 denominator
M3268	Axis 4 servo error reset	D15507	
M3282	Axis 5 forward JOG start	D15648	Axis 5 speed change gear ratio 1 numerator
M3283	Axis 5 reverse JOG start	D15649	
M3287	Axis 5 error reset	D15650	Axis 5 speed change gear ratio 1 denominator
M3288	Axis 5 servo error reset	D15651	
		D15654	Axis 5 speed change gear ratio 2 numerator
		D15655	
		D15656	Axis 5 speed change gear ratio 2 denominator
		D15657	

## User devices


Device No.	Content	Device No.	Content
B0	Automatic operation start (GOT)	W30	Dancer roll 1 A/D value filter: [rad/s]
B1	Home position return (GOT)	W32	Dancer roll 2 A/D value filter: [rad/s]
B2	Error reset (GOT)	W34	Rewinding 1 speed compensation gain: x 0.01[%]
B5	Home position return complete lamp	W36	Rewinding 2 speed compensation gain: x 0.01[%]
B6	Error lamp (GOT)	W38	Rewinding 1 film thickness: x 0.1[μm]
B11	Unwinding axis JOG forward (GOT)	W3A	Rewinding 2 film thickness: x 0.1[μm]
B12	Unwinding axis JOG reverse (GOT)	D1000	Axis 1 control mode
B13	Feed axis JOG forward (GOT)	D1010	Axis 2 control mode
B14	Feed axis JOG reverse (GOT)	D1020	Axis 3 control mode
B15	Cutter axis JOG forward (GOT)	D1030	Axis 4 control mode
B16	Cutter axis JOG reverse (GOT)	D1040	Axis 5 control mode
B17	Rewinding axis 1 JOG forward (GOT)	D2000	Tension detection A/D value
B18	Rewinding axis 1 JOG reverse (GOT)	D2001	Tension detection A/D value conversion: x 0.1[N]
B19	Rewinding axis 2 JOG forward (GOT)	D2002	Tension deviation: x 0.1[N]
B1A	Rewinding axis 2 JOG reverse (GOT)	D2004	Tension compensation value: x 0.01[%]
M1	Axis 1 control mode switching request	D2006 to D2009	Tension deviation cumulative value
M2	Axis 2 control mode switching request	D2020 to D2023	Tension detection A/D filter output value
M3	Axis 3 control mode switching request	D2030	Dancer roll 1 A/D value
M4	Axis 4 control mode switching request	D2031	Dancer roll 2 A/D value
M5	Axis 5 control mode switching request	D2040 to D2043	Dancer roll 1 A/D filter output value
W0	Unwinding axis JOG speed setting (GOT): x 0.01[mm/min]	D2050 to D2053	Dancer roll 2 A/D filter output value
W1	Feed axis JOG speed setting (GOT): x 0.01[mm/min]		
W2	Cutter axis JOG speed setting (GOT): x 0.01[mm/min]		
W3	Rewinding axis JOG speed setting (GOT): x 0.01[mm/min]		
W4	Line speed setting: x 0.01[mm/min]		
W5	Tension setting: x 0.1[N]		
W6	Tension compensation gain: x 0.0001[%/N]		
W7	Tension compensation upper limit value: x 0.01[%]		
W10	Tension compensation lower limit value: x 0.01[%]		
W20	Tension compensation integration time: [ms]		
W24	Tension detection A/D value filter: [rad/s]		

## [Content of Motion SFC sample program]

### Program configuration


No.	Program Name	Automatic Start	Execution Task	Operation Summary
0	Main	Yes	Normal	Main process
1	Motion Control	Yes	Normal	Motion control
2	Home Position	No	Normal	Home position return
3	JOG operation	No	Normal	JOG operation
4	Auto Operation	No	Normal	Automatic operation
5	Unwinding	No	Event (0.88ms)	Unwinding control
6	Rewinding	No	Event (0.88ms)	Rewinding control

(1) No.0 Main: Main Process Normal Task [Automatic Start]  
 Sets initial data and continuously executes the process.




(2) No.1 Motion control: Motion Control Normal Task [Automatic Start]

Via the switches on the GOT screen each Motion control task can be started.



(3) No.2 Home Position Return Normal Task

Executes the home position return servo program for each axis.



(4) No.3 JOG Operation Normal Task  
Executes the JOG operation for each axis.



When GOT JOG switch is ON, the corresponding JOG command bit turns ON.


[GOT]

W0L: Unwinding axis JOG speed setting  
W2L: Feed axis JOG speed setting  
W4L: Cutter axis JOG speed setting  
W6L: Rewinding axis JOG speed setting

B11: Unwinding axis JOG forward  
B12: Unwinding axis JOG reverse  
B13: Feed axis JOG forward  
B14: Feed axis JOG reverse  
B15: Cutter axis JOG forward  
B16: Cutter axis JOG reverse  
B17: Rewinding axis 1 JOG forward  
B18: Rewinding axis 1 JOG reverse  
B19: Rewinding axis 2 JOG forward  
B1A: Rewinding axis 2 JOG reverse

(5) No.4 Auto Operation: Automatic Operation Normal Task

Change to speed control mode in synchronous control to start the unwinding or rewinding task, and then start the command generation axis (main shaft for synchronization).



[G 13]

//////////Control Mode Change//////////

//From Speed/Torque mode to Position mode//

```
//Setting Control Mode
D1000=0 //Ax. 1:Position mode
D1010=0 //Ax. 2:Position mode
D1020=0 //Ax. 3:Position mode
D1030=0 //Ax. 4:Position mode
D1040=0 //Ax. 5:Position mode

//Control Mode Switching Request
SET M1 //Ax. 1
SET M2 //Ax. 2
SET M3 //Ax. 3
SET M4 //Ax. 4
SET M5 //Ax. 5

//Confirmation Servo Status [Control Mode]
(#8010&HC)==H0)*(#8030&HC)==H0)*(#8050&HC)
==H0)*(#8070&HC)==H0)*(#8090&HC)==H0)
```

Axis 1 to 5 Position control mode switching

[F 6]

//Control Mode Switching Request OFF

```
RST M1 //Ax. 1
RST M2 //Ax. 2
RST M3 //Ax. 3
RST M4 //Ax. 4
RST M5 //Ax. 5
```

[G 14]

//Synchronous control OFF

```
RST M12000//Ax. 1
RST M12001//Ax. 2
RST M12002//Ax. 3
RST M12003//Ax. 4
RST M12004//Ax. 5

//Synchronous control OFF Check
!M10880!*M10881!*M10882!*M10883!*M10884
```

Axis 1 to 5 Synchronous control stop

END

## (6) No.5 Unwinding Control Event Task (0.88ms)

Executes the speed control by tension detector on the unwinding axis.

```

Unwinding

[F 4]
///Initial Data Set///
D2006F=0//Tension dev. integration
D2010L=D2L//Unwinding Previous value
D2012L=D22L//Roller Previous value
D15050L=18000000//Initial circumference setting

[G 10]
///////////
// Unwinding Tension Control
///////////

//// F/B Tension detection value //
//
FROM D2000, H0, K11, K1//A/D Ch. 1 value Read
D2020F=W2C/(2.0*3.14)*(0.88/1000.0)*(D2000-D2020F)+D2020F//Filter calc.

D2001=SHORT(RND(D2020F/2.0))//A/D Tension Conversion[x0.1N]
//D2000:A/D value (A/D:0 to 200.0[N] / 0 to 10[V] / 0 to 4000)
//D2020F: A/D Low-pass filter value
//W2C:Low-pass filter ω(Default:5[rad/s])
//

//// Tension PI calculation //
//
D2002=D2001-W20//Tension deviation [x0.1N]
D2004=SHORT(RND((0.001*W22)*(D2006F+D2002)))//Correction value [x0.01%]

IF D2004>W24//Upper limit over
D2004=W24//Upper value
IEND

IF D2004<W26//Lower limit over
D2004=W26//Lower value
IEND

IF (W26<=D2004)*(D2004<=W24)//Limit check OK
D2006F=D2006F+D2002*(0.88/W28)//Tension accumulated deviation
D15054L=10000+D2004///Ax.1 Speed change gear1:Numerator
IEND

//D2002:Tension deviation[x0.1N]
//D2004:Speed Correction value[x0.01%]
//D2006F:Tension accumulated deviation
//W20:Target tension (Default 1000 : 100.0[N])
//W22:Tension P gain (Default 50 : 0.005[%/N])
//W24:Tension Correction upper (Default 500 : 5.00[%])
//W26:Tension Correction lower (Default -500 : -5.00[%])
//W28:Tension Integration time (Default 5000 : 5000[ms])

///////// Roll dia. calculation ///////////
//Roller Circumference:600.0000[mm] φ:190.9859[mm]

IF (D2L-D2010L)>=6000000//Unwinding roll one revolution
D15050L=D22L-D2012L//Roller feed value (Film roll Circumference)

D2010L=D2L//Unwinding Previous value
D2012L=D22L//Roller Previous value
IEND

//D2L:Ax.1 Real current value [x0.1μm]
//D22L:Ax.2 Real current value [x0.1μm]
//D2010L:Unwinding Previous value [x0.1μm]
//D2012L:Roller Previous value [x0.1μm]
//D15050L:Film Roll Circumference [x0.1μm] (gear2:Denominator)

//Auto OFF
!B0

```

### Setting for unwinding control start

- Deviation cumulative value clear for PI operation
- Previous value initial setting for film outer circumference operation
- Set initial value for film outer circumference

### Input for feedback tension

- Tension detector A/D value reading
- Low-pass filter operation
- A/D value → Tension[N] conversion

### Operation for speed compensation value

- Tension deviation operation
- Speed compensation value operation (PI operation)
- Compensation value upper/lower limit check
- Deviation cumulative value operation
- Operation result → Speed gear ratio numerator conversion

### Film outer circumference operation

- Feed axis movement amount detection for unwinding axis per revolution.
- Film outer circumference value → Change gear ratio denominator
- Current value memory for unwinding/rewinding axis. (previous value)

## (7) No.6 Rewinding Control Event Task (0.88ms)

Executes the speed control by the dancer roll on the rewinding axis.

```

    Rewinding
[F_5]
///Initial Data Set///
D2032L=D62L//Rewinding1 Previous value
D2034L=D82L//Rewinding2 Previous value

//Rewinding Film Roll initial dia.
D15500L=1920000//Rewinding 1 :192.0000[mm]
D15650L=1920000//Rewinding 2 :192.0000[mm]

[G_12]
////////// Rewinding Speed Control
////////// Dancer Roll Position ///////////
//A/D:-4000 to 4000/-10V to 10V
//Upper position :10V [4000]
//Reference position :0V [0]
//Lower position :-10V [-4000]

FROM D2030,H0,K12,K2//A/D Ch.2,3 value Read
D2040F=W30/(2.0*3.14)*(0.88/1000.0)*(D2030-D2040F)+D2040F//D.Roll 1 Filter calc.
D2050F=W32/(2.0*3.14)*(0.88/1000.0)*(D2031-D2050F)+D2050F//D.Roll 2 Filter calc.

//Dancer roll 1 (Rewinding 1)//// //Dancer roll 2 (Rewinding 2)////
//D2030:Dancer roll A/D value //D2031:Dancer roll A/D value
//D2040F:A/D Low-pass filter value //D2050F:A/D Low-pass filter value
//W30:Low-pass filter ω(Default:5[rad/s]) //W32:Low-pass filter ω(Default:5[rad/s])
///

////////// Rewinding Speed Compensation ///////////
//Max. Compensation value:±10% (±11000/10000)

//Rewinding 1
D15504L=LONG(RND(10000.0-W34*(D2040F/4000.0)))//Ax.4 Speed change gear2:Numerator
D15506L=10000//Ax.4 Speed change gear2:Denominator

//Rewinding 2
D15654L=LONG(RND(10000.0-W36*(D2050F/4000.0)))//Ax.5 Speed change gear2:Numerator
D15656L=10000//Ax.5 Speed change gear2:Denominator
//W34:Rewinding1 Speed Compensation P gain (Default:10.00[%])
//W36:Rewinding2 Speed Compensation P gain (Default:10.00[%])
////////// Roll dia. Calculation ///////////
//Roller φ:190.9859[mm] Circumference:600.0000[mm]

//Rewinding 1
IF (D80L-D2032L)>=6000000//roll one revolution
  D15498L=1909859 //Ax.4 Speed change gear1:Numerator (Rewinding rollerφ)
  D15500L=D15500L+W38*2 //Ax.4 Speed change gear1:Denominator (Film Roll dia.)
  D2032L=D62L//Rewinding Previous value
IEND

//Rewinding 2
IF (D80L-D2034L)>=6000000//roll one revolution
  D15648L=1909859 //Ax.5 Speed change gear1:Numerator (Rewinding rollerφ)
  D15650L=D15650L+W3A*2 //Ax.5 Speed change gear1:Denominator (Film Roll dia.)
  D2034L=D82L//Rewinding Previous value
IEND

//Rewinding 1 //Rewinding 2
//D2032L: Previous value [x0.1μm] //D2034L: Previous value [x0.1μm]
//D15498L:Ax.4 Speed change gear1:Numerator //D15648L:Ax.5 Speed change gear1:Numerator
//D15500L:Ax.4 Speed change gear1:Denominator //D15650L:Ax.5 Speed change gear1:Denominator
//W38:Film thickness [x0.1μm] //W3A:Film thickness [x0.1μm]

////////// Control Completion ///////////
!B0//Auto OFF

```

END

### Setting for rewinding control start

- Previous value initial setting for film outer diameter operation.
- Initial value setting for film outer diameter

### Input for dancer roll position

- Dancer roll A/D value reading
- Low-pass filter operation

### Operation for speed compensation value

- Compensated value: Proportional gain× A/D value/4000  
10000 - Compensated value  
→ Speed gear ratio numerator.

### Film outer diameter operation

- Film outer diameter for rewinding axis per revolution,  
= Current film diameter + "Film thickness x 2"
- Set film diameter to speed change gear ratio denominator.
- Store current value for rewinding axis.